Shader "UShaderBook/C6/BlinnPhong"
{
Properties
    {
        _Diffuse ("Diffuse", Color) = (1,1,1,1)
        _Specular ("Specular", Color) = (1,1,1,1)
        _Gross ("Gross", Range(8.0, 256.0)) = 20.0
    }
    SubShader
    {
        Pass{
            Tags {"LightMode"="ForwardBase"}
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #include "Lighting.cginc"

            fixed4 _Diffuse;
            fixed4 _Specular;
            float _Gross;

            struct a2v{
                float4 vertex : POSITION;
                float3 normal : NORMAL; 
            };

            struct v2f{
                float4 pos : SV_POSITION;
                fixed3 worldNormal: TEXCOORD0;// 顶点着色器不再需要计算颜色，而是把世界空间法线传给片元着色器
                fixed3 worldPos : TEXCOORD1;
            };

            v2f vert(a2v v){
                v2f o;
                o.pos = UnityObjectToClipPos(v.vertex);
                o.worldNormal = UnityObjectToWorldNormal(v.normal);
                o.worldPos = mul(unity_ObjectToWorld, v.vertex); 
                return o;
            }

            float4 frag(v2f i) : SV_TARGET{
                fixed3 ambient = UNITY_LIGHTMODEL_AMBIENT.xyz; // 环境光颜色获取
                fixed3 worldNormal = normalize(i.worldNormal); // 归一化法线方向
                // 漫反射 diffuse
                fixed3 worldLightDir = normalize(UnityWorldSpaceLightDir(i.worldPos)); // 光源方向
                fixed3 diffuse = _LightColor0.rgb * _Diffuse.rgb * saturate(dot(worldNormal, worldLightDir));
                //高光反射 specular
                fixed3 viewDir = normalize(UnityWorldSpaceViewDir(i.worldPos)); // 顶点到相机的方向
                fixed3 halfDir = normalize(viewDir + worldLightDir); // 中间变量
                fixed3 specular = _LightColor0.rgb * _Specular.rgb * pow(saturate(dot(halfDir, worldNormal)), _Gross);
                fixed3 color = diffuse + ambient + specular;
                return fixed4(color, 1.0);
            }
            ENDCG
        }
    }
    Fallback "Diffuse"
}
